
探索构造法解题模式

【关键字】构造法  数学模型
【摘要】

本文通过一些实例探讨构造法在信息学竞赛解题中的应用，首先阐述了数
学方法在解题中的巧妙应用，引进了数学建模的思想。较详细地讨论建立模型的
方法，包括直接构造问题解答的模型，图论模型，网络流模型以及组合数学模
型。介绍了构建模型的基本方法和基本思路。同时也分析了数学模型的类型和作
用。
【正文】

引言

“构造法”解题，就是构造数学模型解决问题。信息学竞赛中，它的应用十分
广泛。构造恰当的模型或方法，能使问题的解决，变得非常简洁巧妙。
就我们现在所能接触的问题而言，构造的数学模型，从数学方法的分类来

看，它是属初等模型、优化模型这两种。
一般地，数学模型具有三大功能：
1．解释功能：就是用数学模型说明事物发生的原因；
2．判断功能：用数学模型判断原来的知识，认识的可靠性。
3．预见功能：利用数学模型的知识、规律和未来的发展，为人们的行为提

供指导或参考。
构造法解题的思路或步骤可以归纳为：

本文的目的，在于利用构造数学模型的思想，构建我们对问题的解法。

数学的巧妙应用

数学是研究现实世界数量关系和空间形式的科学，数学的特点不仅在于概
念的抽象性、逻辑的严密性、结论的明确性，而且在于它应用的广泛性。我们讲
数学方法是指把错综复杂的问题简化、抽象为合理的数学结构的方法。
我们以具体的问题为例析，解释这些观点的应用，通过这些问题展示了数

学的奇妙作用，让我们体会利用数学方法来解决问题时的一种乐趣。

〖问题 1〗跳棋问题
设有一个 n×n方格的棋盘，布满棋子。跳棋规则如下：

1. 每枚棋子跳动时，其相邻方格（有公共边的方格）必须有一枚棋子为垫子，
才能起跳；

2. 棋子只能沿水平或垂直方向跳动；
3. 棋子跳过垫子进入同一方向的空格，并把垫子取出棋盘。
把 n×n方阵棋盘扩展成m×m，试求出最小的m，使得棋子能依规则跳动，

直到棋盘内只剩下一枚棋子，并给出一种跳棋方案。

本题若用盲目搜索法解决，对 n=4,5或许能行，但也要很高的费用。试用构
造法求解。下面我们把 n按除以 3的余数进行分类讨论：

A. n=3k（k∈N）
我们把 n×n的棋盘分成 k×k个 3×3的网格，并重复排列这种 3×3的网格，

延伸至整个平面。 每个 3×3的网格按下列方式进行着色：

1 2 3

2 3 1

3 1 2

每次跳棋都是从 3色中的两色跳至另一色。设每次跳棋跳至 1、2、3色的次数
分别为 a、b、c次。不妨设最终剩下的一枚棋子是 1色。可得：

3k2 + a – b – c = 1  ①
3k2 – a + b – c = 0  ②
3k2 – a – b + c = 0  ③

① – ②，得 2(a+b)=1，由于 a,b都是整数，本式不成立。所以对于 n=3k的情
况，不可能最终只剩下一枚棋子。

B. n=3k+2（k∈N）
我们先从较小的 n开始，研究是否有规律可循。同时为了对较大的 n能得到

解法的规律，我们构造几种基本形状的跳法。当 n=2时，解法如下：

我们把它定义为基本形状A。
(1) 基本形状 B

(2) 基本形状 C

以上两种基本跳法告诉我们：对于任意的连续 3个棋子，只要另有一个棋
子辅助，就可以连续地消去 3个棋子。有了这几种基本跳法，就可以开始构造本
题的解答了。我们以 n=8为例来说明。如下图所示，其中的正方形表示基本形状
跳法A，竖形长条表示基本形状跳法 C，横形长条表示基本形状跳法 B。其跳法
是：首先从左到右消去左上边的竖形长条，接着从上到下消去右上边的横形长
条，最后从左到右顺序消去下边的正方形与竖形长条。

C. n=3k+1（k∈N）
类似于情况 B，我们仍用基本跳法来构造解答。以 n=7为例，从上到下，从

左到右，逐个消去两种基本形状，最后剩下右下角的一个刀把形，再独立完成。

观察上述过程可以发现，只需把原棋盘向四周扩展 1行，变成(n+2)(n+2)即可。

上题是构造法运用的一个典型例子。构造法不像搜索、动态规划等，有固定
的模式可套用，而完全需要依据实际情况进行状态分析、构图分析对问题进行抽
象性处理，这些，通常需要有良好的数学功底和创造性思维能力，严谨的科学
精神。

〖问题 2〗圆桌吃饭问题
n个人围着一张圆桌吃饭，每个人都不愿意两天与同一人为邻，问最多能

坐多少天，并给出一种排列方案？
为了清楚的理解问题的实质，我们以图的模型来描述它：设 G=(V,E)为一完

全图，|V|=n。图中的每个顶点代表一个人，连结顶点的边表示人之间的相邻关
系。因此，每种围绕圆桌的吃饭方案就成为图中的一条哈密尔顿回路。设
L=<v1,v2,…,vn>为 G中的一条哈密尔顿回路，其中所含的边的集合记为 e(L)。
试求m与 L1,L2,…,Lm，使得 e(Li)∩e(Lj)=φ，并且m达到最大值。
为了求得m的最大值，我们可以先估算一下m上界。完全图G共有 n(n-1)/

2条边，每条哈密尔顿回路含有 n条边，可见m的上界是(n-1)/2。当 n为奇数时
m=(n-1)/2；当 n为偶数时，m=n/2-1。那么这个上界是不是精确上界呢？如能构
造出m条哈密尔顿回路，也就证明了它是本题的答案。
现给出构造方法：作一圆，把圆周分成 n-1等分，标上 n-1个刻度，将顶点

1至 n-1依次排列在圆周上，顶点 n放在圆心。先从圆心出发，向任意点连一条
线，再从这点出发，沿圆周向左右两个方向迂回连线，直到连完圆周上所有的
点，再连回圆心。这样就构造出一条哈密尔顿回路。保持所有的顶点位置不变，
把所有连线围绕圆心逆时针方向旋转一个刻度，得到一条新的哈密尔顿回路。这
样连续旋转(n-1)div 2次，就得到了(n-1)div 2条回路。
下面来证明此算法的正确性。这只要证明所有的边旋转时都不重叠即可。 观

察下图，可以把所有边分为两大类，圆的半径和弦，弦又可以按它的长度分为

(n-1)div 2类。显而易见，不同类别的边在旋转时是不会重叠的。那么相同类别的
边会不会重叠呢？其实每类边至多只有两条，且又处在圆中相对的位置上，所
以当所有的边都旋转半周以后，是不会重叠的。
其算法描述如下：

Procedure Table;
1 k:=0; I:=1; J:=n-1;
2 repeat
3   inc(k);
4   if odd(k)
5     then  r[k]:=I;
6          inc(I);
7     else  r[k]:=J:
8          dec(J);
9 until I>J;
10 for l:=1 to (n-1) div 2 do
11   for i:=1 to k do
12     write((d[i]-1+l)mod(n-1)+1,' ');
13   writeln(n);

〖问题 3〗千足虫问题
千足虫"ishongololo"，身长，色黑而亮，是一种多脚的节肢动物。考虑一个

长为K宽为W，高为 H的各面相互垂直的固体（K,W,H≤32），它从(1,1,1)开始
爬进长方体果实，并会吃光它所经过的路上的一切果实。有下列限定条件：

1．千足虫严格地占据 1个空的小立方块 block。
2．千足虫每次吃完一个小立方块 block。
3．千足虫不能进入以前自己进入过的小立方块 block。
4．千足虫不能进入未吃过的小立方块 block，也不能爬到果实之外。
5．千足虫只能吃掉或只能爬入相邻的 block，此外该 block还必须没有别的

面暴露于已被吃光的 block。
请编程求一个千足虫的动作序列，使它尽可能多地吃掉小立方块。

Toxic解题思路：
本题是 IOI’97的一道试题，作为国际竞赛中的难题，本题有相当的技巧。

首先表现在这是一道三维空间题，需要强的空间想象能力；其次，这题状态较
为复杂，数量巨大。显然，搜索只能作为对小数据作探索性实验时使用的算法。
原题是依据解答的近似程度评分，这又降低了解题的难度。
我们先讨论一种简单的情况，H=1。

 平面的 Toxic路线构造
问题要求在三维长方体中寻找路径。简单起见，我们从平面的 Toxic问题出

发（当MaxZ = 1时）。怎样才能在一个平面矩形中吃到尽可能多的立方体呢？
由于走过的路径不能第二次经过，所以要避免路线过早地将图形分为两部分，

如图 x1的路线就是一个“失败”的例子，区域 A再也无法到达。所以我们考虑基
本路线采用“蜿蜒”前进的策略，如图 x2。

基本路线确定以后，还需要确定两条“主线”之间的间隙。显然，主线之间隔
开 1~2行较好，超过 2行就会有整行的空行无法吃到。图 x3-a与 x3-b是主线间
隔分别为 1、2行时构造出的解。间隔 1行时主线较为密集，但是沿途不能吃其他
的立方体；间隔 2行虽然主线只占总行数的 1/3，但是沿途可以“间隔”地吃立方
体（图 x3-b的浅阴影）。比较发现，间隔 2行能吃到大约 2/3的立方体，间隔 1
行只能吃到大约 1/2的立方体，所以我们采用间隔 2行的策略。

通过和搜索程序的结果比较，在平面上构造出的解和最优解相当接近。
 立体的 Toxic路线构造

有了平面的解，我们再以平面构造的解为基础，构造立体的千足虫路线。
平面构造的解是否可以直接套用在空间长方体上呢？如果千足虫在每一层

都能象图 x3-b一样移动，那么解应当是相当优的。遗憾的是千足虫在某一层爬
过以后，会使得相邻的层有一些立方体无法吃到。准确地说，当立方体(x, y, z)
被吃掉以后，如果千足虫继续再 Z轴坐标为 z的平面内移动，那么(x, y, z+1)以
及(x, y, z-1)这两个立方体就都吃不到了。可以看出，吃掉方格影响的相邻层，这
种影响不会“隔层”，也就是说(x, y, z)被吃掉不会影响到 Z轴坐标为 z+i(|i|>1)的
层面。这就启发我们，可以让千足虫在奇数层上以图 x3-b的方式移动，当在奇
数层从一个顶点移动到另一个顶点时，向上走（吃）2个立方体。暂时不考虑偶
数层的吃法，这样各层的移动互不干扰，至少是一个可行解。我们把这种吃法定
为立体 Toxic问题解的基本路线。

如果仅仅在奇数层上移动、吃立方体，那么总共能吃到的立方体应当是全
部的 1/3左右。我们考虑再吃一些立方体。基本路线在每个偶数层只停留（吃
掉）1格，偶数层大部分的立方体没有被吃掉。我们以第 1层为例，考虑偶数层
的一般吃法。由于第 1层吃掉一些立方体，第 1层相应地有若干方格无法吃到。
如图 x4，第 2层的粗虚线部分在第 3层是无法吃到的，细虚线部分有大约 1/2
的部分是无法吃到的。第 2层的立方体只能是千足虫在 1、3两层的主线上移动时
吃。由于到了第 3层已经不会因为吃第 2层的立方体而影响主线的顺利延伸（如
果从第 1层吃第 2层的果实则不然），所以在第 3层应当将第 2层的立方体尽可
能多第吃。这样，就可以在第 2层 1/3的行（对应于第 3层的主线）间隔地吃到
1/2的果实。如果所有的奇数层都向下尽可能多地吃，那么可以多吃到所有立方
体大约 1/12的立方体，现在一共可以吃到大约 1/3 + 1/12 = 5/12的立方体了。
 沿途“拾遗”

在奇数层我们之所以不往向上 1层吃，是担心影响向上 2层主线的延伸。当
所有路线确定以后，我们可以在不影响路线的前提下由奇数层往向上 1层吃
（在基本路线确定以前则不能随便这样吃）。这样，又可以吃到大约总数 1/12
的立方体（略少于 1/12）。实际上，由于基本路线的确定，具体编程时可以在不
影响主线和遵守吃立方体规则的情况下尽可能多地吃立方体。

至此，构造基本完成。我们的方法大约可以吃到所有立方体中的 1/2。
 坐标变换

我们确定的策略以平面为基础。当MaxX=MaxY=MaxZ时，下底面的选择
无关紧要，而当三者不同时，下底面的选择影响到最终吃的立方体数。所以我们
通过坐标变换将 3个不同的底面作为下底面，这样就可以在尽量不增加编程复
杂度的情况下得到更优的解。经过测试，下底面的选择对解的优劣影响在最大情
况下可以达到 0~15%。

本题的编程实现有一定的难度，具体情况详见附录。

几种模式的构建

以上我们讨论了构造法解题的一种类型，即直接构造问题解答，这只是构
造法运用的一种简单类型。它只能针对问题本身，探索其独有性质，不具备可推
广性。如果要进一步探索，得到各种问题的共性，就要把构造法的思想上升到数
学建模的高度。构造法作为一种解题方法，更多地体现的是数学建模的思想。对
于在这部分中讨论的数学模型，我们按照其主要作用分为两大类：

1．认识事物的模型
这类模型是对现实世界的一个抽象。它提取了其中的有效信息，用简明的方

式表达出来。它可以是一条代数公式、一幅几何图形，也可以是一个物理原理、
化学方程式，甚至是一种自然现象。只要是对客观事物起到抽象概括作用，有利
于我们开拓思维的，都可以算是这类模型。

2．指导算法的模型
本类模型通常是经过前人的大量研究，具有丰富、和谐的性质和定理的一门

理论。在信息学竞赛中，较常用的有图论模型、组合数学模型、流网络模型等。这
类模型有各式经典算法可供套用，对我们算法的设计起决定作用。
下面举一些实例来说明各种模型的建立和具体应用。

〖问题 4〗01串问题
给定N、L0、A0、B0、L1、A1、B1，设计一个长度为N的 01串，使得对于任何

连续的长度为 L0的子串，0的个数大等于 A0且小等于 B0，对于任何连续的长度
为 L1的子串，1的个数大等于A1且小等于 B1。
本题是NOI’99中的一道试题，考试时许多同学都是采用深度搜索算法，加

上限界截枝，5个数据可在规定时间内解出 4个正确答案。现在反思本题，是否
存在高效快捷的多项式时间算法呢？首先我们用数学中多元方程与不等式的思
想构造本题的第一个模型。
模型 4.1
设所求的 01串为 S[1..n]，S[i] ∈{‘0’,’1’}，X[i](I∈{0..n})表示 S的前 I个字符

中’1’的个数。要使该数据模型X[0..n]成为原题的解答，还需满足下列条件：
A．它是一个 01串

X[0]  =  0
X[0] ≤ X[1] ≤ X[0]+1
X[1] ≤ X[2] ≤ X[1]+1

                 
X[n-1] ≤ X[n] ≤ X[n-1]+1

B．它满足子串 01个数限制
S的所有长度为 L的子串中’1’的个数为X[i+L]-X[i](0≤i≤n-L)。因此有

  A0 ≤ X[L0] - X[0] ≤ B0  
  A0 ≤ X[L0+1] - X[1] ≤ B0  

                
  A0 ≤ X[n] - X[n- L0] ≤ B0  

L1-B1 ≤ X[L1] - X[0] ≤ L1-B0  
L1-B1 ≤ X[L1+1] - X[1] ≤ L1-B0  

                
L1-B1 ≤ X[n] - X[n- L1] ≤ L1-B0  

求以上不等式组的一组整数解。若无整数解，则原问题也是无解。
模型 4.1是属于上面所说的第一类模型，它用简明扼要的形式化语言重述

了原题。让我们来分析一下这个模型。仔细观察上述条件，发现它有以下特点：
a．除X[0]=0外，其余的条件都是由“≤”联接的不等式；

b．每个不等式都是含有两个未知数，一个常数的一次不等式；
c．可以通过移项，把每个不等式所含的两个未知数分别移到不等号的两边，

并使它们的系数都等 1。
可见，所有的不等式都可以整理成这样的形式

X[i] + C ≤ X[j]   （i、j =0..n）
它给我们以很大的启发，但我们仍然不能从这个模型直接得到高效算法。
上述不等式类似于连结两点的一条有向边。因此，我们联想到信息学解题中

常用的图论知识。
模型 4.2
首先，构造一个有 n+1个顶点的有向图G，把其顶点编号为 0..n。若有不等

式 X[i] + C≤X[j]，则添加一条从点 i指向 j的有向边，其权为 C。以下我们分根
据图G的性质分两种情况进行讨论。

(1) 图G中不含正圈
这时，图的顶点间存在着最长路径。令 D[i]表示从顶点 0到顶点 i的最长路

径长度。对于图中每条从点 i 指向点 j 的权为 C[i,j]有向边，有性质 D[i]
+C[i,j]≤D[j]。这可用反证法证明：若有D[i]+C[i,j]＞D[j]，即从点 0至点 i再到点
j的一条路径长度大于到点 j的最长路径长度，这与 D[j]的定义矛盾。显然，
D[i]=0。这样，让X[i]=D[i]，X完全符合所有的限制条件，即为原不等式组的一
组解。

(2) 图G中含有正圈
设一个正圈为 v1→v2→v3→…→vm→v1，其路径权和为 C＞0。有

       X[v1] + C1 ≤ X[v2] 
       X[v2] + C2 ≤ X[v3] 
              
      X[vm] + Cm ≤ X[v1] 

相加，得 X[1] + C ≤ X[1] 
               C ≤ 0
这与上述假设矛盾。所以此时原不等式组无解。
对于上述第一种情况，可采用动态规划解决。事实上，由于最长路径长度最

大为 n，只要在 n次迭代内未得到最短路径，就可以判定图中含有正圈，也就是
上述第二种情况。算法如下：
procedure 01_string;
1 由不等式构造有 n+1个顶点的有向图G；
2 X[0..n]  0;
3 K0;
4 Repeat
5   Inc(k); 
6   modifyfalse;
7   for I0 to n do
8     for J0 to n do
9       if x[i]+c[i,j]≤x[j] then
10         modifytrue;
11         x[j]x[i]+c[i,j];
12 Until not modify or (k＞n);

13 If k＞n 
14   then Return(无解)
15   else 依据X[0..n]构造 01串;

事实上，本题的 G边数很少，连结每个顶点的至多只有 4条边。所以 G在
存储结构上，不必使用邻接矩阵。可采用下列定义

    G : array [1..n,1..4] of integer;
上述算法中对图的有关操作（如枚举连结顶点的每条边），只要略加修改即可。
综观本题解题的解题过程，包括了两次构造模型。第一次，是在原题的基础

上构造不等式模型；第二次，是在不等式模型的基础上构造图论模型。本题构造
的实质也就是不断把原问题转化为等价的新问题，直到产生出易于求解的问题
为止。

〖问题 5〗整数拆分问题
给定整数N，求把N拆分成 k个小于 n（n＞N/2）的非负整数的方法总数。
该问题实际上是求不定方程 X1+X2+…+Xk=N（其中 0≤Xi≤n-1）的解的个

数。由于只要求解的总数，可考虑构造母函数的的方法。
解状态可以转化为：求多项式 S=(x0+x1+…+xn-1)k中 xN的系数AN。
设多项式G=x0+x1+…+xn-1，G=（1-xn）/(1-x)，所以 S=Gk。由级数

由于 n的值可能比较大，所以 N的值也可能比较大，求解中需要进行高精
度计算。
本题利用母函数法，得到问题的数学模型，再通过数学方法求解，进而只

要进行高精度求组合数就可得到问题的解。

模型与算法

无论是直接构造问题解答，还是构造数学模型，其算法都是我们最关心的
问题。如何设计一个有较低编程复杂度和时空复杂度且结构清晰的算法，是非常
重要的。以下列几点出发来考虑：

1．选择的模型必须尽量多体现问题的本质特征。但这并不意味着模型越复
杂越好，累赘的信息会影响算法的效率。

2．模型的建立不是一个一蹴而就的过程，而要经过反复地检验、修改，在
实践中不断完善。

3．数学模型通常有严格的形式，但编程实现时则可不拘一格。
我们以实例深入讨论。

〖问题 6〗旅馆经营问题
Patiya城是一座美丽的旅游城市，每年假期都有众多游客来到 Patiya度假 。

Royal Rose酒店就坐落在 Patiya的海滨，它共有 R间客房。假期将要来临，酒店
收到了许多各大旅游公司的客房订单，但是毕竟酒店的客房数有限，不能容纳
所有的游客。现在假设你是这家酒店馆的经理，请你有选择地接受一部分订单，
使你的酒店可以从这些订单中得到最大的收益。每份订单给出了它的起止时间和
预定房间数。一旦你接受了某份订单，在订单规定的时间内，你的酒店就必须为
旅客提供整洁的房间，否则你将因为失去信誉而被辞退。

我们首先要在原题的基础上，剔除与解题无关的信息，建立第一类模型。
模型 6.1
建立时间轴，设从第一个旅客进驻酒店，直到最后一个旅客离开饭店，共

为期 n天。把第一天开始的时刻编号为 1，一、二两天分隔的时刻编号为 2，最后
到把最后一天结束的时刻编号为 n+1。把每份订单用时间轴上的一条以这些点为
端点的线段来表示，设以顶点 i,j（i<j）为端点的线段条数为 c[i,j]。要求在这些
线段中选择一部分来覆盖时间轴，使得每段线段的被覆盖次数不超过 R，并且
覆盖的线段总长度最长。

            1     2     3                 n    n+1 
模型 6.1把问题转化为了一个更加直观的形式，状态更加明确了。在此基础

上，我们首先考虑到的是搜索算法。比如说，对每一线段的次数逐一枚举 ,寻找
最优解。盲目搜索是一种基本算法，其解题过程对问题的分析还停留在表层。尽
管搜索还可以做很多优化，但它仍然是一个时间复杂度为指数阶的算法。为了得
到高效算法，对本题的求解做了一个尝试，使用贪心算法求解：
procedure Hotel_2;
1 设置线段集 S为所有覆盖线段的集合;
2 设置线段集A为空集;
3 for I:=1 to R do
4     从 S寻找一组互相不重叠的线段的集合 V，使它们在时间轴上覆盖的长
度之和最长;

5     S  S – V;
6     A  A + V;  
7 输出A;
上述算法的第 4步可以用动态规划实现。用 d[I]表示覆盖(1,I)的一组互相不

重叠的线段的最大长度和。
D[1]=0

c0[I,J]表示在剩余线段集 S中线段(I,J)的次数。
我们对这诱人的算法抱有很大的希望，但是算法的正确性毕竟要接受实践

的检验。暂不证明这个算法正确性（最优性），我们考虑是否能举出一个反例否
定它。

如上图，若采用上述贪心算法，第一次选择的是线段 1和 2，第二次选择的
是线段 4，这样总长度为 12。但实际上本题的最优解是 4条线段都选择，总长度
为 14。可见上述贪心只是一个近似算法。于是我们又将上述算法的第 4步改为：
    从 S寻找一组线段的集合 V，使得 A+V中每条线段的覆盖次数不超过 I，并
且它们在时间轴上覆盖的长度之和最长;
这个算法初看似乎可以对付上述反例，但我们仍可以举出类似的反例否定

它。
经过对贪心算法的探讨，我们发现求解过程中各种数量之间的制约关系较

为复杂，一般的图论模型无法描述。因此我们想到了迄今为止最复杂的“流网络”
的模型。是否能利用“流网络极值”问题的高效算法解决本题呢？尝试用如下方式
构造本题的“流网络”：
模型 6.2
(1) 构造 n+1个顶点(从 1到 n+1编号)，代表时间轴上的 n+1个标度，并给

源和汇分别编号为 0和 n+2；
(2) 从点 I到 I+1(0≤I≤n+1)，画一条容量上限为+∞，费用为 0的弧。
(3) 对每条线段<I,J>（起点和终点分别为 I和 J），画一条从点 I到点 J的弧，

其容量上限为 c[I,J]，费用为它的长度的弧。它的流量表示这条线段的覆盖次数。
对上面构造的流网络求流量为 R的最大费用流，其流量对应的覆盖方式就

是原问题的解。
算法描述：
proceudre Hotel_3;
1 构造流网络；
2 for I:=1 to R do
3   寻找一条具有最大费用的增流路径；
4 输出流网络对应的解答

仍然以上面那个例子来说明求最大费用流的过程。

     0   1    2   3    4   5    6   7    8    9   10   11  12  13

如图，第一次增流的路径为<0,1,6,7,8,12,13>，第二次增流的路径为
<0,1,2，3,4,5,7,6,9,10,11,12,13>，这时的最大费用为 14。通过对求最大费用流过
程的剖析与它和贪心算法的对比，发现了它们的本质区别在于允许<7,6>的逆向
流过程。

总结

构造与建模是一个复杂的抽象过程。我们要善于透视问题的本质，寻找突破
口，进而选择适当的模型。构造的模型可以帮助我们认识问题，不同的模型从不
同的角度反映问题，可以引发不同的思路，起到引导发散思维的作用。但认识问
题的最终目的是解决问题，模型的固有性质可帮我们建立算法，其优劣可以通
过时空复杂度等指标来衡量，但要以最终实现的程序为标准。所以模型不是一成
不变的，同样要通过各种技术不断优化。模型的产生虽然是人脑思维的产物，但
它仍然是客观事物的在人脑中的反映。所以要培养良好的建模能力，还必须靠在
平时的学习中积累丰富的知识和经验。

【参考书目】
算法与数据结构  （傅清祥  王晓东 编著）
信息学（计算机）奥林匹克入门  （江文哉主编）
信息学奥林匹克杂志

论文附录

【附录 1】论文中引用的原题

〖问题 3〗千足虫

                          暗的千足虫（ishongololo）

    祖鲁族称千足虫为"ishongololo"，它身长，色黑而亮，是一种多脚的节肢动物．千足虫会
吃光它所经过的路上的一切果实．我们以这种事实为基础来理解本题．让我们考虑一个长
为K宽为W，高为H的各面相互垂直的固体．

    请你编程，在题目限定的条件下，使千足虫尽可能多地吃掉小立方块 block．
程序的输出是千足虫吃掉每个 Block所经路线和动作．千足虫从果实之外开始，吃的第一
个 block必须是(1,1,1)，然后必须再爬到这个 block上直到无路可走或无 block可吃时为止。

限定条件：
1．千足虫严格地占据 1个空的小立方块 block。
2．千足虫每次吃完一个小立方块 block。
3．千足虫不能进入以前自己进入过的小立方块 block。
  （此即不能往回走，也不能跨越自己已经走过的路线）．
4．千足虫不能进入未吃过的小立方块 block，也不能爬到果实之外。
5．千足虫只能吃掉或只能爬入相邻的 block，即该 block与千足虫所在的 block共面。此外
该 block还必须没有别的面暴露于已被吃光的 block。

输入
  你的程序将接收到 3个整数即长度 L，宽度W，和
  高度H．L，W，H是三个整数，每个数占一行，且
  它们的取值范围在 1到 32之间（含 1和 32）．

输出
  输出数据由若干行组成．每行以"E"（表示吃 Eat)或M（表示移动Move）打头，后跟三个
整，这三个整数表示千足虫“吃掉”或“移入其内”的小立方块(block)．

评分标准
    如果千足虫违反了约束条件，那么你的答案只得零分．
    所得总分为吃掉的小立方块的总数和已知最优解之比．
    所得最高分不会超过 1OO％

〖问题 4〗01串

01串



给定 7个整数N,A0,B0,L0,A1,B1,L1，要求设计一个 01串 S=s1s2…si…sN，满足：

1. si=0或 si=1，1<=i<=N；
2. 对于 S的任何连续的长度为 L0的子串 sjsj+1…sj+L0-1(1<=j<=N-L0+1)，0的个数大于

等于A0且小于等于 B0;

3. 对于 S的任何连续的长度为 L1的子串 sjsj+1…sj+L1-1(1<=j<=N-L1+1)，1的个数大于
等于A1且小于等于 B1;

例如，N=6,A0=1,B0=2,L0=3,A1=1,B1=1,L1=2，则存在一个满足上述所有条件的 01串
S=010101。

输入
  仅 一 行 ， 有 7 个 整 数 ， 依 次 表 示 N,A0,B0,L0,A1,B1,L1 （ 3<=N<=1000 ， 1<=

A0<=B0<=L0<=N，1<=A1<=B1<=L1<=N），相邻两个整数之间用一个空格分隔。

输出
  仅一行，若不存在满足所有条件的 01串，则输出一个整数-1，否则输出一个满足所有条
件的 01串。

样例输入
  6 1 2 3 1 1 2

样例输出
  010101

【附录 2】跳棋问题的实现(Tiao.pas)

键盘输入棋盘大小 n，结果输入到屏幕
uses crt;

const

    ch          :    array[0..1]of char = ('.','*');

var

    a           :    array[0..100,0..100]of byte;  {棋盘}

    n,i,j,k,k1,num :    integer;  

procedure Print;  {打印棋盘}

  var ii,jj : integer;

  begin

  clrscr;

  inc(num); writeln('No.',num);

  for ii:=0 to n+1 do

    begin

    for jj:=0 to n+1 do

      write(ch[a[ii,jj]],' ');

    writeln;



    end;

  write('Press <ENTER>...');

  readln;

  end;

procedure l(t1,t2:integer);  {基本跳法 C}

  begin

  a[t1,t2-1]:=1; a[t1,t2]:=0; a[t1,t2+1]:=0;

  print;

  a[t1+1,t2]:=0; a[t1+2,t2]:=0; a[t1,t2]:=1;

  print;

  a[t1,t2-1]:=0; a[t1,t2]:=0; a[t1,t2+1]:=1;

  print;

  end;

procedure h(t1,t2:integer);  {基本跳法 B}

  begin

  a[t1,t2]:=0;a[t1+1,t2]:=0;a[t1-1,t2]:=1;

  print;

  a[t1,t2+1]:=0;a[t1,t2+2]:=0;a[t1,t2]:=1;

  print;

  a[t1-1,t2]:=0;a[t1,t2]:=0;a[t1+1,t2]:=1;

  print;

  end;

procedure Mo2(t1,t2:integer);  {另一种基本跳法}

  begin

  a[t1,t2]:=0;a[t1+1,t2]:=0;a[t1+2,t2]:=1;print;

  a[t1,t2+1]:=0;a[t1+1,t2+1]:=0;a[t1+2,t2+1]:=1;print;

  a[t1+2,t2]:=0;a[t1+2,t2+1]:=0;a[t1+2,t2+2]:=1;print;

  a[t1,t2+2]:=0;a[t1,t2+3]:=0;a[t1,t2+1]:=1;print;

  a[t1+2,t2+2]:=0;a[t1+1,t2+2]:=0;a[t1,t2+2]:=1;print;

  a[t1,t2+1]:=0;a[t1,t2+2]:=0;a[t1,t2+3]:=1;print;

  end;

begin

  clrscr;

  write('N='); readln(n);

  if n mod 3 = 0 then begin write('No Way!');readln;halt;end;  {无解}

  num:=0;

  fillchar(a,sizeof(a),0);

  for i:=1 to n do

    for j:=1 to n do

      a[i,j]:=1;

  print;

  k1:=n div 3;

  if n mod 3 = 1 then dec(k1);

  for i:=1 to k1 do



    for j:=1 to n-3 do

      l((i-1)*3+1,j);

  for i:=1 to k1*3 do h(i,n-2);

  case n mod 3 of  {分类解决}

    1:begin

      for i:=1 to k1+1 do h(3*k1+1,(i-1)*3+1);

      for i:=1 to n-2 do l(3*k1+2,i);

      a[n,n]:=0; a[n,n-1]:=0; a[n,n-2]:=1; print;

      a[n-1,n]:=0; a[n-1,n-1]:=0; a[n-1,n-2]:=1; print;

      a[n,n-2]:=0; a[n-1,n-2]:=0; a[n-2,n-2]:=1; print;

      a[n-3,n]:=0; a[n-2,n]:=0; a[n-1,n]:=1; print;

      a[n-2,n-2]:=0; a[n-2,n-1]:=0; a[n-2,n]:=1; print;

      a[n-1,n]:=0; a[n-2,n]:=0; a[n,n]:=1; print;

      end;

    2:begin

      for i:=1 to k1 do

        mo2(3*k1+1,(i-1)*3+1);

      a[3*k1+1,3*k1+1]:=0;a[3*k1+2,3*k1+1]:=0;a[3*k1,3*k1+1]:=1;print;

      a[3*k1+1,3*k1+2]:=0;a[3*k1+2,3*k1+2]:=0;a[3*k1,3*k1+2]:=1;print;

      a[3*k1,3*k1+2]:=0;a[3*k1,3*k1+2]:=0;a[3*k1,3*k1]:=1;print;

      end;

  end;

end.

【附录 3】01串问题的实现(Sequence.pas)

输入输出格式见原题。
{$A+,B-,D+,E+,F-,G-,I+,L+,N-,O-,P-,Q-,R-,S+,T-,V+,X+,Y+}

{$M 64000,0,655360}

program Sequence;

const

    fin         =       'sequence.in';                                  {输入文件}

    fon         =       'sequence.out';                                 {输出文件}

    maxN        =       1000;                              {01串长度的最大值}

var

    N           :       integer;                                      {01串长度}

    A           :       array[1..2,0..2]of integer;                        {限定数值}

    D           :       array[0..maxN] of integer;                       {最长路径}

    V           :       array[0..maxN]of integer;    {v[I]表示从点 I出发的有向边条数}

E           :       array[0..maxN,1..6,0..1]of integer;       

   {E[I,J,0]表示从点 I出发的第 J条有向边指向的顶点，E[I,J,1]表示这条边的权值}

    i,j,k,l     :       integer;

procedure Input;  {输入数据}



  var f:text;

  begin

  assign(f,fin); reset(f);

  read(f,n);

  for i:=1 to 2 do

    read(f,a[i,1],a[i,2],a[i,0]);

  j:=a[1,1];

  a[1,1]:=a[1,0]-a[1,2];

  a[1,2]:=a[1,0]-j;

  close(f);

  end;

procedure Construct;  {构建图的模型}

  begin

  fillchar(v,sizeof(v),0);

  for i:=1 to n do

    begin

    inc(v[i-1]); inc(v[i]);

    e[i-1,v[i-1],0]:=i;

    e[i-1,v[i-1],1]:=0;

    e[i,v[i],0]:=i-1;

    e[i,v[i],1]:=-1;

    end;

  for i:=1 to 2 do

    for j:=0 to n-a[i,0] do

      begin

      k:=j+a[i,0];

      inc(v[j]); inc(v[k]);

      e[j,v[j],0]:=k;

      e[j,v[j],1]:=a[i,1];

      e[k,v[k],0]:=j;

      e[k,v[k],1]:=-a[i,2];

      end;

  end;

procedure Done;  {输出}

  var f:text;

  begin

  assign(f,fon); rewrite(f);

  if l> n

  then write(f,-1) else

    for i:=1 to n do

      write(f,d[i]-d[i-1]);

  writeln(f); close(f);

  end;

begin



  Input;

  Construct;

  fillchar(d,sizeof(d),0);

  l:=0;  {迭代次数}

  repeat  {求最长路径}

    inc(l); k:=0;

    for i:=0 to n do

      for j:=1 to v[i] do

        if d[i]+e[i,j,1] > d[e[i,j,0]] then

          begin

          d[e[i,j,0]]:=d[i]+e[i,j,1];

          k:=1;

          end;

  until (k=0) or (l>n);

  Done;

end.

【附录 4】旅馆问题的实现(Hotel.pas)

输入格式：第一行是暑期的总天数 n和客房总数。第 2–n行是一个邻接矩阵的上三角
（不包括对角线），表示预定日期为第 I 天开房，第 J 天退房的订单预定的房间总数。
输出格式：第一行是总收益天数。接下来的一个邻接矩阵的上三角（不包括对角线）

表示对于预定日期为第 I 天开房，第 J 天退房的订单，接受的房间总数。
{$A+,B-,D+,E+,F-,G-,I+,L+,N-,O-,P-,Q-,R-,S+,T-,V+,X+,Y+}

{$M 64000,0,655360}

program Hotel;

const

    fin         =       'hotel.in';                                      {输入文件}

    fon         =       'hotel.out';                                    {输出文件}

    maxN        =       100;                                      {最大总天数}

var

N, M        :       integer;                              {总天数和客房总数}

    C           :       longint;                                       {最优解}

    a           :       array[1..maxn,1..maxn]of integer;                 {容量上限}

    {由于构造的流网络可能是一个多重图（连结两点不只一条边），按其表示的意义不
同分为两类}

f           :       array[1..maxn,1..maxn]of integer;             {第一类边流量}

    f1          :       array[1..maxn]of integer;                    {第二类边流量}

    d, p        :       array[1..maxn]of integer;                         {最大费用}

    i,j,k,l     :       integer;                                 

procedure Input;  {输入}

  var fi:text;

  begin



  assign(fi,fin); reset(fi);

  readln(fi,n,m);

  fillchar(a,sizeof(a),0);

  for i:=1 to n-1 do

    begin

    for j:=i+1 to n do

      read(fi,a[i,j]);

    readln(fi);

    end;

  close(fi);

  end;

procedure Out;  {输出}

  var fo:text;

  begin

  assign(fo,fon); rewrite(fo);

  writeln(fo,C);

  for i:=1 to n-1 do

    begin

    for j:=i+1 to n-1 do

      write(fo,f[i,j],' ');

    writeln(fo,f[i,n]);

    end;

  close(fo);

  end;

begin

  input;

  for k:=1 to m do  {m次增流}

    begin

    for j:=2 to n do

      begin d[j]:=-1; p[j]:=0; end;

    p[1]:=1; d[1]:=0;

    repeat  {寻找最大费用路径}

      l:=0;

      for i:=1 to n do

        for j:=1 to n do

         if (p[i]>0) and (f[i,j]<a[i,j]) and (d[i]+(j-i)>d[j]) then

          begin

          d[j]:=d[i]+(j-i);

          p[j]:=i; l:=1;

          end;

      for i:=1 to n-1 do

        if (p[i]>0) and (d[i]>d[i+1]) then



          begin

          d[i+1]:=d[i];

          p[i+1]:=i; l:=1;

          end;

      for i:=1 to n-1 do

        if (p[i+1]>0) and (f1[i]>0) and (d[i+1]>d[i]) then

          begin

          d[i]:=d[i+1];

          p[i]:=i+1; l:=1;

          end;

    until (l=0);

    if p[n] > 0 then  {若存在增流路径}

      begin

      i:=p[n]; j:=n;

      c:=c+d[n];

      repeat

        if abs(d[i]-d[j])=0 then

          if i<j

          then f1[i]:=f1[i]+1

          else f1[j]:=f1[j]-1

        else

          begin

          f[i,j]:=f[i,j]+1;

          f[j,i]:=-f[i,j];

          end;

        j:=i; i:=p[i];

      until j=1;

      end;

    end;

  out;

end.

【附录 4】千足虫问题的实现

输入输出格式见原题。
注：本题附有千足虫的动作模 拟程序（ Tox_Simu.pas）和搜 索 算法的实验程序
（Toxic_s.pas）。
program Toxic_Game;

const

  name1         =       'toxic.in';  {输入文件}

  name2         =       'toxic.out'; {输出文件}

  go            :       array[1..4, 1..2] of shortint {平面上 4个方向的坐标增量}

                =       ((0, 1), (1, 0), (0, -1), (-1, 0));

  six           :       array[1..6, 1..3] of shortint {立方体 6个相邻面的坐标增量}



                =       ((1, 0, 0), (0, 1, 0), (0, 0, 1),

                         (-1, 0, 0), (0, -1, 0), (0, 0, -1));

type

  Tsize         =       array[1..3] of integer;

  Tblocks       =       array[0..33, 0..33, 0..33] of shortint;

var

  size          :       Tsize;   {长方体的长、宽、高}

  StoE, EtoS,                    {StoE[i]由于坐标变换}

  BStoE         :       Tsize;   {BStoE表示 Best Size to Experiment，记录下最优的坐标变换}

  mx, my, mz    :       integer; {maxX、maxY、maxZ}

  ans, sum      :       integer; {ans存放当前最优解，sum存放当前吃掉的立方体数目}

  blocks        :       Tblocks; {存放当前路径的信息}

       {blocks[x,y,z]:  -1表示走过该格  -2表示已经吃掉该格  大等于 0表示当前暴露的面的
数目}

  b2            :       ^TBlocks; {记录路径，为“拾遗”做准备}

  outf          :       text;     {输出文件}

  print         :       boolean;  {寻找最优解答和打印解用同一个过程，print表示是否打印}

procedure init; {文件初始化}

  var f : text;

  begin

  assign(f, name1);

  reset(f);

  readln(f, size[1], size[2], size[3]);

  close(f);

  assign(outf, name2);

  rewrite(outf)

  end;

procedure initblocks; {初始化 blocks}

  var x, y, z : integer;

  begin

  fillchar(blocks, sizeof(blocks), 255);

  for x := 1 to mx do

    for y := 1 to my do

      for z := 1 to mz do

        blocks[x, y, z] := 0

  end;

procedure say(ch : char; x, y, z : integer); {当 print=TRUE时输出一个命令}

  var o : Tsize;

  begin

  if not print then exit;



  o[ EtoS[1] ] := x;

  o[ EtoS[2] ] := y;

  o[ EtoS[3] ] := z;

  writeln(outf, ch, ' ', o[1], ' ', o[2], ' ', o[3])

  end;

function eat_block(x, y, z : integer) : boolean; {吃掉立方体(x,y,z)，返回是否成功}

  var i, xx, yy, zz : integer;

  begin

  if (x = 0) or (x > mx) or

     (y = 0) or (y > my) or

     (z = 0) or (z > mz) or

     (blocks[x, y, z] <> 1)

    then begin eat_block := false; exit end;

  eat_block := true;

  blocks[x, y, z] := -2;

  say('E', x, y, z);

  inc(sum);

  for i := 1 to 6 do

    begin

    xx := x + six[i, 1];

    yy := y + six[i, 2];

    zz := z + six[i, 3];

    if blocks[xx, yy, zz] >=0

      then inc(blocks[xx, yy, zz])

    end

  end;

procedure pick_block(x, y, z : integer); 

{对(x,y,z)拾遗，如果立方体(x,y,z)只有一个面与路径接触，则吃掉(x,y,z)不会影响路径}

  var i, j, xx, yy, zz : integer;

  begin

  if (x = 0) or (x > mx) or

     (y = 0) or (y > my) or

     (z = 0) or (z > mz) or

     (blocks[x, y, z] <> 1)

    then exit;

  j := 0;

  for i := 1 to 6 do

    begin

    xx := x + six[i, 1];

    yy := y + six[i, 2];

    zz := z + six[i, 3];

    if (xx >= 1) and (xx <= mx) and



       (yy >= 1) and (yy <= my) and

       (zz >= 1) and (zz <= mz) and (b2^[xx, yy, zz] = -1)

      then inc(j)

    end;

  if j = 1 then eat_block(x, y, z)

  end;

procedure pick_6(x, y, z : integer); {对 6个方向拾遗}

  var i : integer;

  begin

  if (z > 0) and print

    then for i := 1 to 6 do

           pick_block(x + six[i, 1], y + six[i, 2], z + six[i, 3])

  end;

procedure make_way; {按照构造的方法路径}

  var nx, ny, nz, ns : integer; 

{(nx, ny, nz)表示当前坐标，ns表示当前状态，ns=1表示在主线上移动，ns>1表示在相邻
主线的衔接处}

      ford, upd      : integer;

  procedure odd_plane; {奇数层的移动和吃食}

  var t, i, h : integer;

  begin

  eat_block(nx, ny, nz);

  pick_6(nx, ny, nz-1);

  blocks[nx, ny, nz-1] := -1;

  say('M', nx, ny, nz);

  if ns = 2

    then ns := 4

    else ns := 1;

  if ny = 1

    then upd := 1

    else upd := 3;

  repeat

    case ns of

    1 : begin

        if nx = 1

          then ford := 2

          else ford := 4;

        h := 1;

        while eat_block(nx + go[ford, 1], ny, nz) do

          begin



          inc(h);

          eat_block(nx, ny, nz-1);

          if nz+1 = mz then eat_block(nx, ny, nz+1);

          if (upd <> 1) or (h < mx-1) then eat_block(nx, ny+1, nz);

          if (upd <> 3) or (h < mx-1) then eat_block(nx, ny-1, nz);

          pick_6(nx, ny, nz);

          blocks[nx, ny, nz] := -1;

          nx := nx + go[ford, 1];

          say('M', nx, ny, nz)

          end;

        ns := 2

        end;

 2..4 : begin

        if eat_block(nx, ny + go[upd, 2], nz)

          then begin

               eat_block(nx, ny, nz-1);

               if nz+1 = mz then eat_block(nx, ny, nz+1);

               pick_6(nx, ny, nz);

               blocks[nx, ny, nz] := -1;

               ny := ny + go[upd, 2];

               say('M', nx, ny, nz);

               ns := ns mod 4+1

               end

          else break

        end

    end

  until false

  end;

  procedure even_plane; {偶数层的移动和吃食}

  begin

  if not eat_block(nx, ny, nz) then exit;

  pick_6(nx, ny, nz-1);

  blocks[nx, ny, nz-1] := -1;

  say('M', nx, ny, nz)

  end;

  begin

  {对坐标进行变换}

  sum := 0;

  EtoS[ StoE[1] ] := 1;

  EtoS[ StoE[2] ] := 2;

  EtoS[ StoE[3] ] := 3;

  mx := Size[ EtoS[1] ];



  my := Size[ EtoS[2] ];

  mz := Size[ EtoS[3] ];

  {开始构造}

  initblocks;

  blocks[1, 1, 1] := 1;

  nx := 1; ny := 1; nz := 0; ns := 3;

  for nz := 1 to mz do

    if odd(nz)

      then odd_plane

      else even_plane;

  {更新当前最优解}

  if sum > ans

    then begin ans := sum; b2^ := blocks; BStoE := StoE end

  end;

  {主程序}

  begin

  new(b2);

  init;

  ans := -1; min := maxint;

  print := false;

  for StoE[1] := 1 to 3 do

    for StoE[2] := 1 to 3 do

      for StoE[3] := 1 to 3 do

        if [ StoE[1], StoE[2], StoE[3] ] = [1..3]

          then make_way;

  StoE := BStoE;

  print := true;

  make_way;

  close(outf);

  writeln('Eat = ', sum);

  writeln('Rate = ', sum /mx/my/mz :0 :2)

  end.


